Adaptive model, best response b for the adversary

This file complements the paper “Scheduling with a processing time oracle” by Dufossé, Dürr, Nadal, Trystram and Vásquez.

Number of jobs: n
Short job length: p
Long job length: p + x
Numher of short tested jobs: c
Number of long tested jobs: d
Delay caused by the tests: e

Suppose algorithm decides to switch to the second phase, namely to execute all remaining n-c-d jobs untested.

Adversary chooses a number b of jobs among these which will be long.

This notebook shows how to compute the optimal value for b.

In[11]:=
```
Clear[ALG, OPT, h]
ALG[b_] := p n (n + 1) / 2 + e + x ((n - c) (n - c + 1) - (n - c - b) (n - c - b + 1) + d (d + 1)) / 2;
OPT[b_] := p n (n + 1) / 2 + x (b + d) (b + d + 1) / 2;
h[b_] := FullSimplify[ALG[b] × OPT[b - 1] - ALG[b - 1] × OPT[b]] 2 / x
```

In[15]:=
```
FullSimplify[D[D[h[b], b], b]]
```

Out[15]= $-2\,(1-c+d+n)\,x$

In[16]:=
```
FullSimplify[Solve[h[b] == 0, b]]
```

Out[16]=
$$\Big\{\Big\{b \to -\Big(\Big(2e+2n(1+n)p+(-1+c+d+2d^2)x-nx-\sqrt{\big(4(e+n(1+n)p)^2+4\big(e(-1+c-d+2cd-n-2dn)+n(1+n)\big((-1+c)c+d+d^2+n-2cn+n^2\big)p\big)x+(1+d-c(1+2d)+n+2dn)^2x^2\big)}\Big)\Big/\big(2(1-c+d+n)x\big)\Big)\Big\},$$
$$\Big\{b \to -\Big(\Big(2e+2n(1+n)p+(-1+c+d+2d^2)x-nx+\sqrt{\big(4(e+n(1+n)p)^2+4\big(e(-1+c-d+2cd-n-2dn)+n(1+n)\big((-1+c)c+d+d^2+n-2cn+n^2\big)p\big)x+(1+d-c(1+2d)+n+2dn)^2x^2\big)}\Big)\Big/\big(2(1-c+d+n)x\big)\Big)\Big\}\Big\}$$

In[17]:=
```
rightEnd = FullSimplify[h[n - c - d]]
```

Out[17]= $2ce-2en+np+cnp+dnp+cn^2p+dn^2p-n^3p+(-1+c-n)(c-n)(1+c+d-n)x$

In[23]:= **alternativeForm = −2 e (n − c) − p n (n + 1) (n − c − d − 1) − x (n − c) (n − c + 1) (n − c − d − 1)**

Out[23]= $-2\,e\,(-c+n) - n\,(1+n)\,(-1-c-d+n)\,p - (-c+n)\,(1-c+n)\,(-1-c-d+n)\,x$

In[24]:= **FullSimplify[rightEnd − alternativeForm]**

Out[24]= 0

In[20]:= **FullSimplify[h[0]]**

Out[20]= $n\,(1+n)\,(1-c+n)\,p - d^3\,x + d^2\,(-c+n)\,x + d\,(-2\,e - n\,(1+n)\,p + (1-c+n)\,x)$

For small p and x, h[0] could be negative. Hence b* can be negative.